



Weld your way.

Carola -

Követelmények

**A Carola programkódra
vonatkozó követelmények**

Tartalomjegyzék

1. Bevezetés	3
2. A berendezések általános hatékonysága (OEE)	3
2.1. A tényleges mennyiség és a jó mennyiség meghatározása	3
2.1.1. Mesterképzés	3
2.1.2. Változók	6
2.1.3. Egyszerű komponensprogram	6
2.1.4. Komplex komponens program	7
3. Megjegyzések	9
4. Változók	9

1. Bevezetés

A C-Gate egyes funkcióihoz szükséges, hogy a Carola programból származó információkat a C-Gate címre továbbítsák. A megfelelő funkcióhoz szükséges blokkokat és utasításokat a megfelelő fejezetekben találja.

Ha a `ReadVar` adatszerkezetet vagy a `OPCUAVAR` parancsot más környezetben használja, győződjön meg arról, hogy a használt területek nem fedik egymást. Felhívjuk a figyelmet arra is, hogy a Carola programban használt index nem azonos a webes felületen használt indexszel. Az értékeket a C-Gate a `ReadVar` adatszerkezeten keresztül adja át. A vezérlőben az első index egy és az utolsó 100. A C-Gate -ben az első index 0 és az utolsó 99. A vezérlőben 50-es index alatt tárolt érték a C-Gate -ben 49-es index alatt kérhető le.

Felhívjuk figyelmét, hogy az alábbiakban ismertetett Carola kód példák részben vagy egészben már el lehetnek tárolva a mesterprogramban. Kérjük, hasonlítsa össze a szakaszokat, és szükség esetén igazítsa ki őket. A függvények és változók megkettőzését a Carola értelmező hibaként ismeri fel.

Megjegyzés: A QIROX `WriteVar` változók változásai befolyásolhatják a Carola programsorozatot.

Megjegyzés: Kérjük, használja az Adobe Acrobat Reader programot a fájl megnyitásához, ha a Carola programkódot szeretné másolni.

2. A berendezések általános hatékonysága (OEE)

2.1. A tényleges mennyiség és a jó mennyiség meghatározása

A tényleges mennyiség és a jó mennyiség meghatározásához a következő programrészeket kell tárolni a fő-, illetve az alkatrészprogramban. Fontos, hogy a következő eljárások kizárólag a 94-100-as indextartományt használják, és nem használhatók más célokra. A tartomány a jövőben bővíthet.

2.1.1. Mesterképzés

Az első sorok azért szükségesek, hogy a változókat más programokból lehessen importálni és kiírni. Közvetlenül a `RESTART` utasítás után található.

```
RESTART
! VÁLTOZÓ DEKLARÁCIÓ
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE, COMPVARI, COMPSERI, COMPSTEP, COMPSEAM
```

Ez a blokk egy olyan eljárást határoz meg, amely információt küld a C-Gate címre arról, hogy egy komponens elindul. Ezt megelőzően az alább használt változókat be kell állítani a komponensprogramban.

```
PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(100)
```

```

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('VÁLTOZÓ >COMPTYPE< MÉG NEM ÍRTÁK LE!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEAM,'COMPSEAM')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
ENDP

```

Ez a blokk egy olyan eljárást határoz meg, amely továbbítja a C-Gate címre azt az információt, hogy egy komponens elkészült.

```

! AZ ALKATRÉSZFELVÉTEL VÉGE
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSEAM := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEAM,'COMPSEAM')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')

```

```
OPCUAVAR(0,95, COMPQUAL, 'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Ez a blokk egy olyan függvényt definiál, amely továbbítja az információt a C-Gate címre, hogy egy új varrás kezdődik. A varrat nevét az eljárás meghívása előtt a COMPSEAM változóba kell írni.

```
! VARRÁS MEGKEZDÉSE
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Ez a blokk egy olyan függvényt definiál, amely továbbítja az információt a C-Gate címre, hogy az aktuális varrás befejeződött.

```
! EGY VARRAT VÉGE
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Ez a blokk egy olyan függvényt definiál, amely lekérdezi a PHG-n, hogy az alkatrész rendben van-e vagy sem. Az ellenőrzés a dolgozó vizuális ellenőrzésére támaszkodik, és a további programáramlást addig blokkolja, amíg a bejegyzés meg nem történik. A AZ ALKATRÉSZ OK=1 VAGY NEM OK=0 ? szövegét az Ön igényeihez igazíthatja.

```
! ŐMINSÉGI KOMPONENSEK ROEGZÍTÉSE
PUBLIC PROC COMPQMAN
WREAD ('AZ ALKATRÉSZ OK=1 VAGY NEM OK=0 ?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

Ez a blokk egy olyan függvényt definiál, amely automatikusan (lekérdezés nélkül) OK-ra állítja a komponens minőségét. Ez a funkció akkor használható, ha nem kívánja dinamikusan rögzíteni a berendezés teljes hatékonyságának (OEE) minőségét, de mégis szeretné, hogy a teljes metrika helyesen legyen kiszámítva.

```
! ŐMINSÉGI KOMPONENSEK ROEGZÍTÉSE
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

2.1.2. Változók

A programban a következő változók használhatók.

- **COMPTYPE** : Az alkatrész típusa.
- **COMPVARI** : Az alkatrész változata.
- **COMPSEMI** Az alkatrész sorozatszám. Ha nincs érték megadva, a C-Gate automatikusan létrehoz egy sorozatszámot az időbélyegző alapján.
- **COMPSTEP** : A gyártási lépés neve. Az érték akkor fontos, ha az alkatrészt több állomáson dolgozzák fel.
- **COMPSEAM** : A varrat neve. Egy varrat több szövegrészből is állhat.

2.1.3. Egyszerű komponensprogram

Kérjük, importálja a változókat és eljárásokat a főprogramból.

```
RESTART
```

```
! VÁLTOZÓK ÉS ELJÁRÁSOK IMPORTÁLÁSA
```

```
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEMI,COMPSTEP,COMPSEAM FROM MASTER
```

```
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
```

```
MAIN
```

A programfolyamatban állítsa be legalább a kötelező, és szükség esetén a választható mezőket is. Amint minden mezőt beállítottunk, hívjuk meg a **COMPSTART** eljárást. Ha megvásárolta a C-Gate.QM modult, akkor a varratokat is meghatározhatja, és finomabban strukturálhatja az alkatrészreklót. Ehhez használja a **SEAMSTRT** és a **SEAMEND** eljárásokat. Amint az alkatrész elkészült, és az alkatrészminőséget a Carola programból kívánja beállítani, hívja a **COMPQAUT** vagy a **COMPQMAN** programot a minőségi adatok átviteléhez. Ha megvásárolta a C-Gate.QM modult, akkor az alkatrészminőséget automatikusan a QIROX SD monitorozással határozza meg. A **COMPQAUT** vagy a **COMPQMAN** eljárás meghívása nincs hatással az alkatrészprotokollra.

A komponens befejezéséhez hívja meg a **COMPEND** eljárást. A hívások sorrendje fontos.

Kérjük, vegye figyelembe, hogy a **COMPEND** meghívása és a komponens újraindítása között legfeljebb egy másodpercnek kell eltelnie a **COMPSTART** címen, különben a vezérlőben bekövetkező adatszinkronizálás miatt elveszhet a végjelzés. A tervek szerint ez a korlátozás a robotvezérlő egyik következő szoftververziójában megszűnik.

```
! AZ ALKATRÉSZINFORMÁCIÓK BEÁLLÍTÁSA
```

```
! ŐKOETELEZ ŐMEZ
```

```
COMPTYPE := 'ŐVEZÉRLSZEKRÉNY'
```

```
! OPCIONÁLIS
```

```
! COMPVARI := 'XL'
```

```
! OPCIONÁLIS
```

```
! COMPSEMI := '1234567890'
```

```

! OPCIONÁLIS
! COMPSTEP := 'PADLÓ'

! AZ ALKATRÉSZ KEZDETE
CALL COMPSTART

! *****
! KOMPONENS PROGRAM
! *****

COMPSEAM := 'HEGESZTÉSI VARRAT 1'
CALL SEAMSTR

! *****
! PROGRAM HEGESZTÉSI VARRAT 1. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.
! *****

COMPSEAM := 'HEGESZTÉSI VARRAT 2'
CALL SEAMSTR

! *****
! PROGRAM HEGESZTÉSI VARRAT 2. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.
! *****

CALL SEAMEND

! A KOMPONENS ÓMINSÉGÉNEK MEGHATÁROZÁSA A CAROLA PROGRAMBAN.
! HÍVJON, ÓMIELTT COMPEND
CALL COMPQMAN
! VAGY
! CALL COMPQAUT

! AZ ALKATRÉSZ VÉGE
! HÍVJA A COMPQMAN VAGY COMPQAUT
CALL COMPEND

! A KOMPONENSPROGRAM VÉGE
END

```

2.1.4. Komplex komponens program

A következő Carola program egy két gyártási lépéssel rendelkező alkatrészt hoz létre. A gyártási lépések hegeszthetők egymás után egy roboton vagy párhuzamosan több roboton. Ennek előfeltétele, hogy a lépések egy protokollban legyenek kombinálva:

- A típusnak, a változatnak és a sorozatszámnak minden robotnál azonosnak kell lennie.
- A megfelelő komponenssablonban be kell állítani a megfelelő minimális gyártási lépések számát. Ebben az esetben két gyártási lépés van hegesztve, ezért a számot legalább a kettő értékre kell beállítani.

```

RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! *****
! GYÁRTÁSI LÉPÉS 1
! *****

COMPTYPE := 'KAZÁN'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
COMPSTEP := 'PADLÓ'

CALL COMPSTART

COMPSEAM := 'HEGESZTÉSI VARRAT 1'
CALL SEAMSTRT
! PROGRAM HEGESZTÉSI VARRAT 1. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.

COMPSEAM := 'HEGESZTÉSI VARRAT 2'
CALL SEAMSTRT
! PROGRAM HEGESZTÉSI VARRAT 2. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! *****
! 2. GYÁRTÁSI LÉPÉS
! *****

COMPTYPE := 'KAZÁN'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
COMPSTEP := 'BORÍTÓ'

CALL COMPSTART

COMPSEAM := 'HEGESZTÉSI VARRAT 3'
CALL SEAMSTRT

```



```
! PROGRAM HEGESZTÉSI VARRAT 3. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.
```

```
COMPSEAM := 'HEGESZTÉSI VARRAT 4'
```

```
CALL SEAMSTRT
```

```
! PROGRAM HEGESZTÉSI VARRAT 4. PÉLDÁUL ÓKUELOENBOEZ GC MOZGALMAK.
```

```
CALL SEAMEND
```

```
CALL COMPQAUT
```

```
CALL COMPEND
```

```
END
```

3. Megjegyzések

Ha Ön PDM- vagy UMS-ügyfél, vagy volt, és a két opció egyike aktiválva van a vezérlő konfigurációjában, akkor a következő megjegyzést sürgősen be kell tartani! Ha a két opció egyike aktiválva van, de a robotvezérlő nincs csatlakoztatva a megfelelő PC-alkalmazáshoz, akkor az adatfolyamban akár 5 másodpercig tartó hézagok keletkeznek, amelyek jelentősen meghamisítják az alkatrészprotokollt. Kérjük, vagy csatlakoztassa a robotvezérlőt a PC alkalmazáshoz, vagy távolítsa el a megfelelő opciót a konfigurációból.

Ha egyszerre több robottal dolgozik ugyanazon az alkatrészen, és szeretné, ha az összes adatot egyetlen alkatrészprotokollban egyesítenék a C-Gate.QM modulon belül, kérjük, további információért forduljon a CLOOS szervizhez. Ellenkező esetben kérjük, hagyja üresen a `COMPSERI` mezőt, mivel ez nemkívánatos mellékhatásokhoz vezethet a berendezések teljes hatékonyságának (OEE) kiszámításában.

4. Változók

A Carola programban lehetőség van változókat csatolni az alkatrészprotokollhoz. Ehhez használja az `OPCUAVAR` parancsot (lásd a példát). Ha a név (utolsó paraméter) a `'COMPVAR_'` előtaggal rendelkezik, akkor a változót a komponenshez csatolja. Ha a név a `"STEPVAR_"` előtaggal rendelkezik, akkor a változó az alkatrész lépéséhez kapcsolódik. Ha a név a `"SEAMVAR_"` előtaggal rendelkezik, a változó az aktuális varrathoz kapcsolódik. Minden kiválasztott névhez csak egy érték tárolható. Az adott névvel utoljára beállított utolsó érték kerül a naplóba. A név előtagja nem kerül tárolásra. Az értéket a varrat, lépés vagy komponens befejezése előtt vissza kell állítani. Ha ez nem történik meg, az értéket a következő elemhez is hozzácsatolja a rendszer. Egész számok, vesszőszámok és szövegek menthetők.

```
CALL SEAMSTRT
```

```
OPCUAVAR(0,1, 'EXAMPLE-VALUE', 'COMPVAR_SEAM_MEASURE_A')
```

```
OPCUAVAR(0,1, '', '')
```

```
CALL SEAMEND
```